Mathematica(ver11)+SIR理論 による Covid-19 のデータの分析

§3．微分方程式で 未来を見る

テーマはあくまでも「*Mathematica*によるデータ分析の試み」です．*Mathematica* に対しても Covid-19 に関しても専門家ではないので，間違いがあるかもしれません．^_^;

2020年5月13日(5月19日改定)．　生越 茂樹．oh.shigeki@gmail.com

0. SIR理論

1. R_0の値を変えて 微分方程式を解く

前章で述べたように γ=0.05(回復期間20日)とします．
γ の値はすぐ下の行で簡単に変えられます．　しかしそもそも(#3)から「時間単位を1/γ に変えると方程式はRの値のみで決まる」ので，
1つのγでだけチェックすれば十分と言えます．　またデフォルトでは対象国は日本で，s,i,r などの初期値は5月8日現在のデータ．
データを変える場合は，下のプログラムの「caseLatest,deathLatest,recLatest, population」を変更して下さい．
(Mathematica notebookの場合．cdfでは固定．しかし 後述してますが，R≥1.1 の時は 初期値には殆ど無関係です．)
ちなみに R_0=2 からのスタートです．

```
DynamicModule[
 {sol, contentR, contents, contenti, contentr, content1, myrange, caseLatest, deathLatest, recLatest, population},
 Manipulate[Quiet[
   sol = NDSolve[{Rt'[t] == Rt0 * s'[t], s'[t] == -γ * Rt[t] * i[t], i'[t] == γ * (Rt[t] - 1) * i[t], r'[t] == γ * i[t],
      Rt[0] == Rt0,
      s[0] == (population - caseLatest) / population,
      i[0] == (caseLatest - recLatest) / population,
      r[0] == recLatest / population}, {Rt, s, i, r}, {t, 1, n}];
   contentR = If[showR && ! ionly, Evaluate[Rt[t] /. sol]];
   contents = If[shows && ! ionly, Evaluate[s[t] /. sol]];
   contenti = If[showi && ! ionly, Evaluate[i[t] /. sol], If[ionly, population * Evaluate[i[t] /. sol]]];
   contentr = If[showr && ! ionly, Evaluate[r[t] /. sol]];
   content1 = If[! ionly, 1];
   myrange = If[ionly, All, Automatic];
   Plot[
    {content1, Tooltip[contentR, "再生産数 R_t"], Tooltip[contentr, "回復者&死者"], Tooltip[contents, "感染可能な人"],
     Tooltip[contenti, "感染者"]}, {t, 0, n}, PlotStyle -> {Dashed, Line, Line, Line, Line},
    PlotLegends -> {"y=1", "再生産数 R_t", "r(回復者&死者)", "s(感染可能な人)", "i(感染者)"},
    ImageSize -> 600, PlotRange -> myrange]],

  {{ionly, False, "i のみを人数表示"}, {True, False}, ControlType -> RadioButton}, Delimiter,
  {{showR, True, "R_t を表示"}, {True, False}}, {{shows, True, "s (人口比) を表示"}, {True, False}},
  {{showi, True, "i (人口比) を表示"}, {True, False}}, {{showr, True, "r (人口比) を表示"}, {True, False}},
  {{Rt0, 2, "R_0"}, 0.5, 5, 0.01, Appearance -> "Labeled"},
  {{n, Floor[20 / γ], "期間"}, Floor[5 / γ], Floor[200 / γ], 1, Appearance -> "Labeled"},
  Initialization :> (
    γ = 0.05;
    caseLatest = 15575; (*この後3つの変数は日本の5月8日現在のデータ*)
    deathLatest = 590;
    recLatest = 5146;
    population = 127484450
   ), FrameLabel -> "R_0の変化 と s,i,rの増減 "]]
```

i のみを人数表示 ○ True ◉ False

R_t を表示 ☑

s (人口比) を表示 ☑

i (人口比) を表示 ☑

r (人口比) を表示 ☑

R_0 2

期間 400



R_0の変化 と s,i,rの増減

■ 2. R_0>1のとき,「感染者の最大値」を求める

$R_t=\frac{S\beta}{\gamma}$ と定義すると (これから $R_0=\frac{N\beta}{\gamma}$, $R_t= R_0 * \frac{S}{N} = R_0 * s$ となる),

$$\frac{ds}{dt}=-R_t*i*\gamma\cdots(1),\quad \frac{di}{dt}=(R_t*i-i)*\gamma=(R_t-1)i*\gamma\cdots(2),\quad \frac{dr}{dt}=i*\gamma\cdots(3)$$

この式はs, i, r が全て人口に対する比率のときも成り立つ. (1),(2)より

$$di/ds=-1+\frac{1}{R_0}\times\frac{1}{s}$$

これを積分して,
$$i+s-\frac{1}{R_0}\log s=一定$$

「t=0の時の i,s,rの値をi_0,s_0,r_0 とすると, $i+s-\frac{1}{R_0}\log s=i_0+s_0-\frac{1}{R_0}\log s_0\cdots(4)$

(2)より R_t >1 のとき, i が最大⇔R_t=1 (グラフからも読み取れる)であり,「$R_t = R_0$ s 」だから,

$$s_{(iが最大の時)}=\frac{1}{R_0}$$

(4)とあわせて i の最大値は,

$$i_{Max}=1-\frac{1}{R_0}(1+\log R_0)+\delta \quad \left(\delta=-r_0-\frac{1}{R_0}\log s_0\right)$$

初期の段階では s_0=1, r_0=0 と考えて良いから, その近似値は

$$M(R_0)=1-\frac{1}{R_0}(1+\log R_0)$$

```
GraphicsRow[{Plot[{Tooltip[(1 - 1 / r (1 + Log[r])), "M(R0)"], Tooltip[Log[r] / r^2, "M'(R0)"]}, {r, 1, 5},
   PlotLabel → "R>1の時, M(R0)=1-(1+logR0)/R0(青) と, その微分(緑)",
   AxesLabel → {"R0", "M"}, PlotLegends → {}, AxesOrigin → {0, 0}], Plot[Evaluate[
    Table[Tooltip[Exp[0.05 (r - 1) t], "R0=" <> ToString[r]], {r, 1, 0.3, -0.1}]], {t, 0, 5 / γ}, PlotRange → {0, 1},
   AxesLabel → {"t 日後", Style["i(t)/i(0)", Larger]}, PlotLabel → "R0≤1 の時の i の変化 (R=0.3~1)"]}, ImageSize → 700]
```



例えば 現在(5月13日), 私の計算では 世界の R_0は 1.3〜1.4 でした.

M(1)=0, M(1.1)=0.004 , M(1.2)=0.015, M(1.3)=0.029, M(1.4)=0.045

```
Clear[r]
```

```
Manipulate[Round[1 - (1 + Log[r]) / r, 0.001], {{r, 2, " R0 "}, 1, 5, 0.01, Appearance → "Labeled"}, FrameLabel → "M(R0)の値"]
```



また 世界人口は77億人, Confirmed Cases(感染者積算)=460万, 死者は30万, 回復者(死者含まず) は 170万, なので,

i_0=0.00034, r_0=0.00026, s_0=0.9994

故に 1<R_0 の時は, δ<0.00075 です. 従って 1.1≤R_0 の時は, δ は無視できます.

$M(R_0)$の値から, 感染者は

R_0=1.4 のとき 3.5億人, R_0=1.3 のとき 2.2億人, R_0=1.2 のとき 1.1億人, R_0=1.1 のとき 0.3億人

仮に致死率2%とすると, 死者は

R_0=1.4 のとき 700万人, R_0=1.3 のとき 400万人, R_0=1.2 のとき 200万人, R_0=1.1 のとき 65万人

$M(R_0)$ は $\sqrt{e} \doteq 1.648$ で極大となるので, $R_0 = 1.65$ 前後では, R_0 の削減の効果が特に大きいです.

ですから, 世界全体にとっては今が頑張りどきと言えるでしょう.

一方 $R_0 < 1$ の時は i は減少関数なので, t = 0 のときiは最小です. そしてこの時 s ≐ 1 のままだから $R_t \doteq R_0$ で (2) より,

$$\frac{di}{dt}=(R_0-1)\,i*\gamma$$

故に　　　　　$i[t] = i[0] * \exp[\gamma(R_0 - 1)\, t]$ （右上図）

そして 6 割の削減で R = 1, 7 割の削減で R = 0.75, 8 割の削減で R = 0.5 となり，確かに8割の削減では急速に収束します．
しかし 以上はあくまでもSIR理論による理論値です．

3. 果たして SIR理論は正しいのか？